text: # put this file somewhere in your $fpath
  # and call below lines
  #  % autoload -Uz anyframe-init
  #  % anyframe-init
  
  local dir file
  for dir in ${^fpath}/anyframe-functions/{actions,selectors,sources}(N-/); do
    fpath=($fpath $dir)
  
    for file in $dir/*(N-.); do
      local function_name="${file:t}"
      autoload -Uz -- "$function_name"
    done
  done
  
  for dir in ${^fpath}/anyframe-functions/widgets(N-/); do
    fpath=($fpath $dir)
  
    for file in $dir/*(N-.); do
      local function_name="${file:t}"
      autoload -Uz -- "$function_name"
      zle -N -- "$function_name"
    done
  done
  
  # Local Variables:
  # mode: Shell-Script
  # End:
  # vim: ft=zsh
  